﻿// Upgrade NOTE: replaced 'mul(UNITY_MATRIX_MVP,*)' with 'UnityObjectToClipPos(*)'

// Upgrade NOTE: replaced '_World2Object' with 'unity_WorldToObject'

Shader "ShaderLearn/LightModeHalfLabert"
{
	Properties
	{
		_Diffuse ("Diffuse", Color) = (1,1,1,1)
	}
	SubShader
	{
		Tags { "LightMode"="ForwardBase" }
		LOD 100

		Pass
		{
			CGPROGRAM
			#pragma vertex vert
			#pragma fragment frag
			
			#include "UnityLightingCommon.cginc"
			#include "UnityCG.cginc"

			struct a2v
			{
				float4 vertex : POSITION;
				float4 normal : NORMAL;
			};

			struct v2f
			{
				float4 vertex : SV_POSITION;
				fixed3 world_normal : TEXCOORD0;
			};

			fixed4 _Diffuse;
			
			v2f vert(a2v v)
			{
				v2f o;
				o.vertex = UnityObjectToClipPos(v.vertex);
				//通过内置函数获得法向量，normalize(mul(v.normal, (float3x3)unity_WorldToObject))
				o.world_normal = UnityObjectToWorldNormal(v.normal);
				return o;
			}
			
			fixed4 frag (v2f i) : SV_Target
			{
				fixed3 ambient = UNITY_LIGHTMODEL_AMBIENT.xyz;	//环境光
				fixed3 worldLight = normalize(_WorldSpaceLightPos0.xyz);
				// 用(color + 1)^3/8处理散射光颜色，相比(color + 1)/2的半兰伯特光照，可以让阴影分界更明显
				fixed3 halfLabert = pow(dot(i.world_normal, worldLight) + 1, 3) / 8;				
				fixed3 diffuse = _LightColor0.rgb * _Diffuse.rgb * halfLabert;
				
				fixed3 color = ambient + diffuse;
				return fixed4(color, 1.0);
			}
			ENDCG
		}
	}
}
